

B e s c h r e i b u n g
des ZUSE-Formel-Codes für die ZUSE Z 23
(Kleiner Formelübersetzer Ausgabe August 1962)

Der Freiburger Code für die ZUSE Z 22 und ZUSE Z 23 gilt allgemein als einer der anschaulichsten Maschinencodes, in denen man mit Rechenanlagen verkehren kann. Trotzdem wird häufig nach einem Code gefragt, der mit den direkten mathematischen Formeln arbeitet.

Aus diesem Grunde wurde für die ZUSE Z 23 ein "Kleiner Formelübersetzer" aufgestellt, der diesen Wünschen weitgehend entgegenkommt. Der ZUSE-Formel-Code ist nicht so umfangreich und allgemein wie ALGOL. Da ALGOL ja auch für den Verkehr zwischen verschiedenen Maschinen vorgesehen ist, muß diese Sprache ganz allgemein gehalten sein, während der ZUSE-Formel-Code nur für den Benutzer der ZUSE Z 23 gedacht ist.

Zur Bezeichnung von Funktionen, Sprüngen, Schleifen wurden Abkürzungen aus der deutschen Sprache benutzt, da die Anlage bisher fast nur im deutschsprachigen Raum eingesetzt wird und man damit dem Ingenieur oder Techniker eine leicht verständliche und erlernbare Programmierungssprache in die Hand gibt.

Der Formelübersetzer verarbeitet einfache Formeln ohne Klammerausdrücke mit Variablen oder Funktionsbildung, kann Entscheidungen treffen, Schleifen bilden u.a. Die Variablen können in einfacher Form oder indiziert auftreten.

Bevor die einzelnen Möglichkeiten erklärt werden, sollen einige Beispiele gegeben werden, wie ein Programm in dieser Formelsprache aussieht.

Beispiel_1 Bildung von $z = \sqrt{a^2 + b^2}$

wobei verschiedene Wertepaare a und b eingelesen werden sollen.

1 } Vorbereitung
x } Bedeutung siehe unter
"Ablochvorschrift"

1) LIES A }
LIES B } Einlesen der Größen a und b

DRU A }
DRU B } Drucken der Größen a und b

C = A . A Bildung von $a^2 = c$
D = B . B " " $b^2 = d$
D = D + C " " $d = d+c = a^2 + b^2$

Z = WURZ D Ziehen der Quadratwurzel

DRU Z Drucken von Z

ZLN 1 1mal Zeilenvorschub

SPR 1 Springen auf die mit der Marke 1
bezeichneten Stelle

START 1 Starten an der mit der Marke 1
bezeichneten Stelle

3,78 a
5,91 b
7,843 a
9,586 b

usw.

Beispiel_2

Für den Wert x von 0 mit der Schrittweite 0,125 bis 4 soll
der Wert der Funktion

$$f = \frac{\sin x}{e^x} \cdot x$$

errechnet werden.

1 } Vorbereitung siehe unter
x } "Ablochvorschrift"

```
1) FUER X = 0 : 0,125 : 4      Eröffnung der Schleife
A = SIN X
B = EXP X
C = A/B
F = C . X
DRU X
DRU F
ZLN 1
WDH X                          Ende der Schleife
STOP
START 1
```

Erklärungen der einzelnen Möglich-
keiten des ZUSE-Formel-Codes

1. Variable

A) Gleitendes Komma-Variable: A, B, . . . Z
(jeweils 1 Buchstabe)

B) Ganzzahlige- und Festkomma-Variable, auch Indizes: A', B'
. . . Z' (jeweils 1 Buchstabe mit Apostroph)

2. Indizierte Variable

Die Variablen in gleitendem und festem Komma können indiziert werden. (Der Index läuft stets ab 0 und kann nur positiv sein)

A) Der Index kann numerisch angegeben werden,

z.B. wird für die Gl.K.V.
(Gleitkomma-Variable) a_7 A7'
für die F.K.V.
(Festkomma-Variable) b_{17} B'17'

geschrieben.

Der Index steht hinter der Variablen-Bezeichnung und ist mit einem Apostroph versehen.

B) Symbolisch angegebener Index

a) einfacher Index

z.B. für Gl.K.V. a_i AI'
" F.K.V. b_k B'K'

b) zusammengesetzter Index

Ein zusammengesetzter Index wird stets in Klammern angegeben und hat die allgemeinste Gestalt

$$A(\overset{+}{I}' \cdot \overset{+}{J}' \pm \overset{+}{L}') \quad (\text{für die Gl.K.V. } a_{\pm n \cdot j \pm i})$$

Es können hierbei einzelne Teile fehlen oder numerisch angegeben werden,

z.B. für a_{3j+k} $A(3 \cdot J' + K')$
 a_{i+1} $A(I' + L')$
 a_{n-1} $A(N' - 1')$

Es ist zu beachten, daß der resultierende Index nur positiv sein darf.

C) Mehrbuchstabile Variablenbezeichnung

Um gewisse Variable besonders hervorzuheben oder um mehr Variablenbezeichnungen zu erhalten, können für Gleitkomma-variable auch mehrbuchstabile Namen eingeführt werden. Diese Namen müssen in eckigen Klammern [] angegeben werden und können aus 1 bis 6 Zeichen bestehen (wobei Bu, Zi, leer und Zw unberücksichtigt bleiben und Wr und Zl nicht erlaubt sind).

Diese Sondernamen können nicht indiziert werden.

z.B. [ALPHA]
[B1]

Für die indizierten Variablen und die mit längeren Namen belegten Variablen müssen in der Vorbereitung Platzreservierungen vorgenommen werden (vergl. Ablochvorschrift).

3. Arithmetische Operationen

$A = B+C$	Addition
$A = B-C$	Subtraktion
$A = B.C$	Multiplikation
$A = B/C$	Division

Hier können A, B, C einfache oder indizierte Variablen (in Gleitkomma- oder Festkomma-Darstellung) sein.

Die beiden rechts stehenden Größen (hier B und C) müssen beide Gl.K.V. oder F.K.V. sein, dagegen können links vom Gleichheitszeichen eine Festkomma-Variable und rechts zwei Gleitkomma-Größen stehen,

$$\text{z.B. } A' = B \cdot C$$

Das bedeutet:

Der ganzzahlige Teil des Produktes der Gleitkomma-Variablen B und C wird als Festkomma-Variable A' gespeichert.

Die rechts des Gleichheitszeichens stehenden Größen können auch numerisch angegeben werden,

$$\text{z.B. } A = 3/B$$

Dem 1. Operanden kann ein Minuszeichen vorangesetzt werden, wobei allerdings bei Addition und Subtraktion das vorangestellte Minuszeichen auf das Ergebnis angewendet wird,

$$\begin{aligned} \text{d.h. } A &= - B+C \\ \text{steht für } a &= - (b+c) \end{aligned}$$

Rechts des Gleichheitszeichens kann auch nur eine Größe stehen, deren Wert wird dann für die links angegebene Variable eingesetzt,

$$\begin{aligned} \text{z.B. } A &= B & A' &= 4' \\ A &= -C & B' &= -6' \end{aligned}$$

4. Funktion

Es sind folgende Funktionen vorgesehen, die nur auf eine Gl.K.V. anwendbar sind.

SIN X	sin x	
COS X	cos x	
TAN X	tg x	
CTG X	ctg x	
ARCSIN X	arc sin x	
ARCTAN X	arc tg x	
EXP X	e^x	
LOG X	ln x	
BETR X	x	
WURZ X	\sqrt{x}	
GANZ X	[x]	ganzzahliger Teil von x
MOD X	x(mod 1)	

Als Gleichung ist jeweils zu schreiben.

$$\begin{aligned} \text{z.B. } Z &= \text{SIN X} \\ Y &= \text{GANZ R} \end{aligned}$$

Den Funktionen als auch den Argumenten kann ein Minuszeichen vorangestellt werden,

$$\begin{aligned} \text{z.B. } Z &= -\text{EXP -X} \\ \text{steht für } z &= -e^{-X} \end{aligned}$$

5. Marken

Vor jede Formel oder sonstige Anweisung (s.u.) kann eine Marke gesetzt werden als Bezeichnung einer Stelle, auf die gesprungen werden soll. Die Marken sind numerisch anzugeben und mit einem) direkt nach der Zahl zu versehen; die folgende Formel ist von der Marke durch einen Zwischenraum zu trennen.

$$\text{z.B. } 2) A = B/C$$

6. Sprünge

A) unbedingte Sprünge:

SPR 7 : Sprung auf die mit der Marke 7 versehene Formel.

B) bedingte Sprünge:

```
WENN A-B 3 7 6
d.h. wenn a - b < 0, springe auf Marke 3
          a - b = 0, " " " 7
          a - b > 0, " " " 6
```

Hierbei kann B auch numerisch angegeben werden:

```
WENN A-3,74 3 7 6
```

Anstelle des -Zeichens kann auch + . oder / stehen.
Es kann auch nur eine Größe nach dem WENN stehen:

```
WENN A 3 7 6
d.h. wenn a < 0, springe auf Marke 3
          a = 0, " " " 7
          a > 0, " " " 6
```

C) Schalterbedingte Sprünge

Um das Programm während des Ablaufes von außen beeinflussen zu können, kann der Bedingungsschalter am Bedienungspult vom Programm abgefragt werden.

Z.B.

```
BED 3:7 5:8
```

heißt, falls der Bedingungsschalter auf 3 steht, springe nach Marke 7, falls er auf 5 steht, springe nach Marke 8, sonst fahre im Programm fort.

Hinter dem Symbol BED kann eine oder mehrere Angaben stehen. Durch Übergang auf eine neue Zeile (vergl. Ablochvorschrift) wird das Schlußzeichen für BED gegeben.

D) Sprung auf Unterprogramm

```
UPR 13
```

Sprung auf das Unterprogramm, das an der Marke 13 beginnt. Der Rücksprung (RSP) aus dem Unterprogramm erfolgt auf die Stelle nach dem Sprung ins Unterprogramm.

Durch

RSP wird das Ende des Unterprogramms bezeichnet
(Rücksprung),

z.B. Produkt der beiden komplexen Größen z_1 und z_2 , um z_3 zu erhalten.

$$z_3 = z_1 \cdot z_2$$

$$\text{wobei } z_1 = x_1 + iy_1$$

$$z_2 = x_2 + iy_2$$

$$z_3 = x_3 + iy_3$$

UPR7

·
·
·

$$7) R = X1' \cdot X2'$$

$$S = Y1' \cdot Y2'$$

$$X3' = R - S$$

$$R = X1' \cdot Y2'$$

$$S = Y1' \cdot X2''$$

$$Y3' = R + S$$

RSP

7. Zyklen (maximal 12 Zyklen ineinander geschachtelt)

a) FUER A = B : C : D

d.h.: Die Gl.K.V. A soll vom Anfangswert B mit der Schrittweite C bis D variiert werden.

$$a = b$$

$$b + c$$

$$b + 2c$$

$$b + 3c$$

$$\text{bis } a = b + nc \leq d$$

Hierbei darf die Schrittweite nur positive Werte haben.
Der Zyklus, in dem A auf diese Weise variiert werden soll,
wird mit WDH A abgeschlossen.

Z.B. Summe aller ganzen Zahlen von 1 bis 100

S = 0

FUER Z = 1 : 1 : 100

S = S + Z

WDH Z

b) Entsprechend können Zyklen für Indizes eingeleitet werden.

FUER I' = A' : B' : C'

Der Zyklus soll für den Index

A'

A' + B'

A' + 2B'

⋮

⋮

⋮

A' + nB' ≤ C'

durchlaufen werden.

Die Formeln, die mit dem veränderlichen Index I' durchlaufen werden, sind direkt unter dem FUER zu schreiben und mit WDH I' abzuschließen.

Die Größen rechts vom Gleichheitszeichen (B : C : B bzw. A' : B' : C') können mit Variablen, indizierten Variablen oder numerisch angegeben werden.

8) Text

Für Überschriften, Erläuterungen usw. kann Text an beliebiger Stelle eingefügt werden. An diesen Stellen ist zu lochen:

TEXT ; ; ☒

der zwischen den ; eingeschlossene Text wird im Programmablauf an dieser Stelle ausgeschrieben. (; ist gleichbedeutend mit ☒ = Klingelzeichen).

9) Einlesen

Durch die Anweisung

LIES A bzw. LIES I'

wird ein Wort vom Lochstreifen eingelesen und an die durch A bzw. I' bezeichnete Stelle gebracht. Die eingelesenen Größen können auch indiziert sein. Ablochvorschrift siehe unten.

10) Ausgabe

Die Ausgabe errechneter Werte kann in verschiedener Form geschehen.

Bei der Druckanweisung ist die Form jeweils mit anzugeben.

a) Ausgabe in Gleitkommaform

DRU A 9'

Der Zahlenwert A soll im gleitenden Komma mit 9stelliger Mantisse ausgegeben werden (z.B. -,314159265/+01)

Hier kann die Anzahl der Stellen auch symbolisch angegeben werden. DRU A N'

b) Ausgabe in Normalform (bei den Gleitkommazahlen wird der Exponent durch Kommaverschiebung berücksichtigt).

DRU B 3' 4'

Der Zahlenwert B soll mit 3 Stellen vor und 4 Stellen nach dem Komma ausgeschrieben werden (z.B. -146,1359).

Hierbei kann die 2. Angabe auch 0 sein, muß aber gemacht werden.

c) Ausgabe von ganzzahligen Festkommazahlen

DRU I' 4'

Die Festkommazahl I' soll 4stellig ausgegeben werden (z.B. -6149')

Falls keine Angabe über die Druckform gemacht ist, wird die Zahl in der Form ausgegeben, wie zuletzt ausgegeben wurde. Wenn überhaupt keine Angabe gemacht wird, so erscheinen die Gleitkommazahlen im gleitenden Komma mit 9 Stellen und die ganzzahligen Festkommazahlen mit 4 Stellen.

11. Tabellen-Form

Soll der nächste Wert auf die nächste Zeile geschrieben werden, so ist zu programmieren:

ZLN 1 einmal Wagen- und Zeilentransport;
 oder mehrfach Zeilentransport:
ZLN 5 5mal Zeilentransport

Nach jeder Zahl werden 2 Zwischenräume freigelassen. Will man mehr Zwischenräume haben bzw. etwas einrücken, so kann man dies durch

ZWI n angeben, wobei n die Anzahl der zu machenden
Zwischenräume angibt.

12. Start und Stop

Durch STOP kann an jeder beliebigen Stelle ein Zwischenstop in den Programmablauf eingeführt werden.

Durch Betätigen der Starttaste wird das Programm fortgesetzt.

Am Ende des Programmstreifens ist

START n

zu geben, wobei n die Marke bezeichnet, an der der Programmablauf beginnen soll.

13. Test

TEST ist verboten, wenn im Programm BED benutzt wird. Von der Stelle an, an der TEST steht, wird, sofern Schalter 14 auf 1 geschaltet ist, für jede Formel ein zusätzlicher Druckbefehl eingeführt.

Das Ausdrucken während des Programms erfolgt jedoch nur, sofern Schalter 14 auf "aus" steht. Es wird das Ergebnis der Formel dann jeweils auf einer Zeile mit 9 Stellen im gleitenden Komma ausgegeben.

Steht Schalter 14 auf 1, wird nur der gewöhnliche Programmablauf durchgeführt.

Auf diese Weise ist es möglich, bei Neuaufrstellung eines Programms ein direktes Protokoll zu bekommen. Ist das Programm getestet, kann man es ohne die zusätzlichen Druckbefehle aufbauen lassen, indem man beim Einlesen Schalter 14 ausschaltet.

Als weitere Testmöglichkeit kann man beim Einlesen Schalter 14 auf 2 stellen. Dann wird für jeden bedingten Sprung (WENN) ein Druckbefehl eingeführt, der die Marke mit folgendem ")" ausschreibt, auf die gesprungen wird.

Ablochvorschrift

Für den Programmstreifen ist abzulochen:

- 1) ein Gleichheitszeichen als Anfangszeichen,
- 2) danach ist die größte im Programm auftretende Marke anzugeben
- 3) Falls mehrbuchstabile Variablen verwendet werden, sind diese hier aufzulisten, jeder Name in eckigen Klammern.
- 4) Falls indizierte Variablen vorkommen, so sind diese danach mit ihrem höchsten Index aufzuführen. Der Index muß hier numerisch gegeben sein. (Es werden dadurch Plätze für die indizierten Variablen vom Index 0 bis zum angegebenen Index reserviert)
- 5) ein \times -Zeichen als Abschlußzeichen der Vorgaben (\times ist identisch mit ?),
- 6) das Programm, wobei jede Formel auf eine Zeile zu schreiben ist. Zwischen Funktionsbezeichnung und Variablen ist jeweils ein Zwischenraum zu lochen. Auch sind die Marken von den folgenden Formeln oder Anweisungen durch Zwischenraum zu trennen.
- 7) Gleitkommazahlen, die in den Formeln direkt auftreten, müssen in der Normalform abgelocht werden. z.B.

3 3.14 0.768

wobei ein Zwischenraum oder Wagen- oder Zeilentransport das Schlußzeichen für die Zahl sind.

8) Festkommazahlen, die direkt in die Formel eingehen oder die auf dem Datenstreifen stehen, können nur ganzzahlig sein und müssen mit einem Apostroph direkt hinter der letzten Ziffer versehen sein.

Der Programmstreifen wird mit Eo+1 eingelesen.

Der Datenstreifen

9) Auf dem Datenstreifen können Gleitkomma-Variable in der Normalform oder in Gleitkommadarstellung gelocht werden, wobei der Dezimal-Exponent durch einen Schrägstrich von der Mantisse getrennt wird, z.B.

3	7,895	0,0071	Normalform
68,76/+10	für die Zahl	687600000000	

Abschließend noch einige Bemerkungen zum Umfang des Übersetzerprogramms.

Der Übersetzer wird in die Zellen ab 1040 eingespeichert und benötigt etwa 850 Speicherplätze, die Funktionen werden anschließend eingespeichert und belegen 412 Speicherplätze.

Das übersetzte Programm wird ab Zelle 2400 eingespeichert. Für Programm und Speicherraum für die Daten stehen 5 600 Speicherplätze zur Verfügung. (Für eine Anweisung werden im Übersetzerprogramm durchschnittlich 3 Speicherzellen benötigt). Dieser Speicherraum kann noch vergrößert werden, indem man zunächst nur den Übersetzer einliest, nicht aber die Funktion und das übersetzte Programm ab Zelle 1988 einspeichern läßt. Bevor der Ablauf des Programms beginnt, werden dann die Funktionen angefordert durch den Text

BITTE UNTERPROGRAMME

Nach Einlesen der Unterprogramme kann dann mit Buchstaben J vom Lochstreifen her der Ablauf des Programms gestartet werden (oder in Trommelzelle 1040).

Falls die Funktionen

SIN, COS, TAN, CTG, ARCSIN, ARCTAN, EXP, LOG, GANZ und MOD

nicht benötigt werden, kann auch nach Ausdrucken des Textes

BITTE UNTERPROGRAMME

mit J bzw. 1040 gestartet werden. (WURZ und BETR sind stets vorhanden).

Die Einteilung, ab wo das übersetzte Programm eingespeichert wird, übernimmt die Maschine selbst.

Beispiele

$$1) \quad Z = c \cdot \sqrt{\frac{P_n(x)}{P_m(x)}} = c \cdot \sqrt{\frac{\sum_{i=0}^n a_i x^{n-i}}{\sum_{i=0}^m b_i x^{m-i}}}$$

Für den Wert x von 0 mit der Schrittweite 0,25 bis 5,0 soll der Wert der Funktion errechnet werden.

=
 6'
 A20' B20' Anzahl der gelieferten Werte
 X

```

1) LIES N'
   FUER I' = 0':1':N'
   LIES AI'
   WDH I'
   LIES M'
   FUER I' = 0':1':M'
   LIES BI'
   WDH I'
   LIES C
   FUER X = 0:0,25:5
   DRU X 1' 2'
   D = A0'
   FUER I' = 1':1':N'
   D = D * X
   D = D + AI'
   WDH I'

```

```
E = B0'  
FUER I' = 1':1':M'  
E = E . X  
E = E + BI'  
WDH I'  
WENN E 3 2 3  
2) TEXT ; UNENDLICH; ✱  
SPR 6  
3) D = D / E  
WENN D 4 5 5  
4) TEXT ; IMAGINAER; ✱  
SPR 6  
5) Z = WURZ D  
Z = C . Z  
DRU Z 9'  
6) ZLN 1  
WDH X  
STOP  
  
START 1
```

D = Ergebnis des Zählerpolynoms

E = Ergebnis des Nennerpolynoms

I = Index der Zyklen; da 4 Zyklen nacheinander ablaufen,
konnte für alle der gleiche Index benutzt werden.

Die Polynomkoeffizienten müssen in der Reihenfolge

n' : $a_0, a_1, a_2 \dots a_n$,

m'' : $b_0, b_1, b_2 \dots b_m$,

c

abgelocht sein.

$n', m'' \leq 20'$: Falls mehr Koeffizienten vorhanden sind, muß
die Angabe A20' B20' entsprechend geändert werden.

Mit den Werten

5'

2,5 1,43 0,2 3,12 2,0 1,25

4'

5,7 3,25 1,87 0,56 2,75

0,545

erhält man folgende Ergebnisse

```

0      , 367438924/-00
0,25   , 434338161/-00
0,50   , 474010722/-00
0,75   , 475701536/-00
1,00   , 469807635/-00
1,25   , 473865160/-00
1,50   , 487915232/-00
1,75   , 508088675/-00
2,00   , 531390558/-00
2,25   , 556052197/-00
2,50   , 581107500/-00
2,75   , 606044239/-00
3,00   , 630597859/-00
3,25   , 654638573/-00
3,50   , 678110311/-00
3,75   , 700997389/-00
4,00   , 723306096/-00
4,25   , 745054506/-00
4,50   , 766266337/-00
4,75   , 786967776/-00
5,00   , 807185530/-00
    
```

2) U.P. Integration einer durch Unterprogramm gegebenen Funktion $f(x)$ im Intervall a, b nach der Simpson'schen Regel.

Aufruf mit UPR 1

```

1) F:=0
2) R:= B-A
   N:=2 * N
   D:=R/N
   M:=0
   S2' :=0
   X:=A
   UPR 9
   S1' :=Y
   W:=-1,0
3) M:=M+1,0
   R:=D * M
   X:=R+A
   UPR 9
   WENN W 4 5 5
4) R:=4 * Y
   S3' :=R+S1'
   W=1,0
   SPR 3
5) S1' :=Y
   S2' :=Y+S2''
   S2' :=S2'+S3''
   W:=-1,0
   WENN M-N 3 6 6
6) R:=D/3
   S3'' :=R * S2''
    
```



```
R=S3"-F
R=R/S3"
R=BETR R
WENN R-0,00001 8 7 7
7) F=S3'
   SPR 2
8) RSP
```

A = Anfang des Intervalls
B = Ende des Intervalls
N = Anzahl der Teilintervalle
F = berechnete Fläche
R = Zwischenspeicher
D = Intervallbreite $\frac{b-a}{2n}$
M = Zählung der Teilintervalle
x = Argument für die gegebene Funktion
y = f(x)

S1 = y_i
S2 = $y_i + 4 y_{i+1}$
S3 = \sum
W = Weiche


Bei Aufruf des U.P. Integration müssen die Parameter a, b und n als Gleitkommazahlen in den Speichern A, B und N bereitgestellt sein.

Als "Vorbereitung" ist vom Hauptprogramm aus zu setzen:

höchste Marke: mindestens 9
Variable: S3"
Marke 9: Beginn des U.P. zur Berechnung von f(x)

z.B. 9) Y=X.X
Y=Y.0,75
Y=Y+3,0
RSP

$$y = \frac{3}{4}x^2 + 3$$

10) A=0
B=4,0
N=4
UPR 1
TEXT ;F"; 
DRU F 3" 3"
STOP
START 10

Ergebnis: F = 28.000

3) Skalarprodukt zweier n-dimensionaler Vektoren.

Der Aufruf dieses Unterprogramms erfolgt mit UPR 5

Die benötigten Eingabedaten müssen in der Reihenfolge

$$\begin{array}{ccccccc} n' & & & & & & \\ x_1' & x_2' & x_3' & \cdot & \cdot & \cdot & x_n' \\ y_1' & y_2' & y_3' & \cdot & \cdot & \cdot & y_n' \end{array}$$

abgelocht sein.

Das Ausdrucken des Ergebnisses erfolgt vom Hauptprogramm aus, da es im Unterprogramm nur gespeichert wird.

```
5
X10' Y10'
X
1) UPR 5
   DRU S 3' 3'
   ZLN 1
   STOP

5) LIES N'
   FUER I' = 1':1':N'
   LIES XI'
   WDH I'
   FUER I' = 1':1':N'
   LIES YI'
   WDH I'
   S=0
   FUER I' = 1':1':N'
   A = XI' * YI'
   S=S+A
   WDH I'
   RSP

START 1
```

Wenn die Dimension n der Vektoren > 10 ist, muß im Hauptprogramm die Angabe X10' Y10' entsprechend geändert werden.

Bad Hersfeld, 12.9.1962

Su/Bk

*Schnellspeicher und
Variable Zuse-Formel-Code.*

A	201	A'	227
B	202	B'	228
C	203	C'	229
D	204	D'	230
E	205	E'	231
F	206	F'	232
G	207	G'	233
H	208	H'	234
I	209	I'	235
J	210	J'	236
K	211	K'	237
L	212	L'	238
M	224	M'	250
N	214	N'	240
O	215	O'	241
P	216	P'	242
Q	217	Q'	243
R	218	R'	244
S	219	S'	245
T	220	T'	246
U	221	U'	247
V	222	V'	248
W	223	W'	249
X	213	X'	239
Y	225	Y'	251
Z	226	Z'	252